Differential Geoemetry

Homework 3

Jackson Henry

2/16/2015

```
ConvertMetric[OldMetric_, OldCoords_, NewCoords_] :=
  Table[Sum[D[OldCoords[[i]], a] D[OldCoords[[j]], b] OldMetric[[i, j]],
    {i, Length[OldCoords]}, {j, Length[OldCoords]}], {a, NewCoords}, {b, NewCoords}];
```

```
Γ[c_, a_, b_, met_, coord_] := Module[{imet, n}, n = Length[coord]; imet = Inverse[met];
Return[(1 / 2) Sum[imet[[c, d]] (D[met[[a, d]], coord[[b]]] +
        D[met[[b, d]], coord[[a]]] - D[met[[a, b]], coord[[d]]]), {d, 1, n}]]]
(* R[μ_,ν_,ρ_,σ_,met_,coord_] gives R_abc^d for the given metric *)
R[μ_, ν_, ρ_, σ_, met_, coord_] :=
 D[Γ[σ, μ, ρ, met, coord], coord[[ν]]] - D[Γ[σ, ν, ρ, met, coord], coord[[μ]]] +
  Sum[Γ[α, μ, ρ, met, coord] Γ[σ, α, ν, met, coord] - Γ[α, ν, ρ, met, coord] Γ[σ, α, μ, met, coord],
   {α, 1, Length[coord]}]
(* R[μ_,ν_,met_,coord_] gives the Ricci tensorR_ab for the given metric *)
R[μ_, ρ_, met_, coord_] := Sum[R[μ, ν, ρ, ν, met, coord], {ν, 1, Length[coord]}]
(* R[met_,coord_] gives the Ricci Scalar R*)
R[met_, coord_] := Module[{imet, n}, n = Length[coord]; imet = Inverse[met];
  Return[Sum[imet[[a, b]] R[a, b, met, coord], {a, 1, n}, {b, 1, n}]]]
```

Problem 1

```
NewCoords = {x, y};
OldCoords = {x, y, Sqrt[r^2 - x^2 - y^2]};
OldMetric = {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
NewMetric = ConvertMetric[OldMetric, OldCoords, NewCoords];
RicciScalar = Simplify[R[NewMetric, NewCoords]]
```

$$\frac{2}{r^2}$$

Problem 2

```
NewCoords = {t, ϕ};
OldCoords = {t - Tanh[t], Sech[t] Cos[ϕ], Sech[t] Sin[ϕ]};
OldMetric = {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
NewMetric = ConvertMetric[OldMetric, OldCoords, NewCoords];
RicciScalar = Simplify[R[NewMetric, NewCoords]]
ParametricPlot3D[OldCoords, {t, 0, 3}, {ϕ, 0, 2 Pi}]
```

```
-2
```



Problem 3

```
Solve[x^2/F^2 + y^2/G^2 + z^2/H^2 == 1, z]
```

$$\left\{\left\{z \to -\frac{H\sqrt{F^2 G^2 - G^2 x^2 - F^2 y^2}}{F G}\right\}, \left\{z \to \frac{H\sqrt{F^2 G^2 - G^2 x^2 - F^2 y^2}}{F G}\right\}\right\}$$

```
NewCoords = {x, y};
OldCoords = {x, y, H Sqrt[F^2 G^2 - G^2 x^2 - F^2 y^2]/(F G)};
OldMetric = {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
NewMetric = ConvertMetric[OldMetric, OldCoords, NewCoords];
RicciScalar = Simplify[R[NewMetric, NewCoords]]
```

$$\frac{2 F^6 G^6 H^2}{\left(-F^2 G^4 x^2 + G^4 H^2 x^2 + F^4 \left(G^4 - G^2 y^2 + H^2 y^2\right)\right)^2}$$